Macaulay2で不変式環の計算

安田が2024年度前期に担当している講義「代数幾何学特論1」の、2024年6月11日の講義用資料。

参考ウェブページ

- Macaulay2: https://macaulay2.com/
- Macaulay2 documentation: https://macaulay2.com/doc/Macaulay2/share/doc/Macaulay2/Macaulay2Doc/html/
- InvariantRingパッケージの説明：https://macaulay2.com/doc/Macaulay2/share/doc/Macaulay2/InvariantRing/html/index.html

まず簡単な例を計算してみる

まず起動する。

```
+ M2 --no-readline --print-width 94
Macaulay2, version 1.24.05
with packages: ConwayPolynomials, Elimination, IntegralClosure, Inverse
Systems, Isomorphism,
               LLLBases, MinimalPrimes, OnlineLookup, PrimaryDecomposit
ion, ReesAlgebra,
               Saturation, TangentCone, Truncations, Varieties
```

次に不変式環を計算するためのパッケージを読み込む。

```
i1 : loadPackage "InvariantRing"

o1 = InvariantRing

o1 : Package
```

このパッケージは線形簡約群の線形作用に対する不変式環を計算するためのパッケージ。有限群で標数と位数が互いに素(tame case)な場合はOK。

ここで、まず簡単な不変式環を一つ計算してみよう。最初に有理数係数の2変数多項式環を用意する。

```
i2 : R = QQ[x,y]

o2 = R
```

```
o2 : PolynomialRing
```

次に、`x`と`y`の互換に対応する行列を用意し、その行列により定まる多項式環`R`への群作用を定義する。

```
i3 : M = permutationMatrix [2,1]

o3 = | 0 1 |
     | 1 0 |

              2         2
o3 : Matrix ZZ  <--- ZZ

i4 : myAction = finiteAction(M,R)

o4 = R <- {| 0 1 |}
           | 1 0 |

o4 : FiniteGroupAction
```

不変式環の生成元を求めるには、コマンド`invariants`を使う。

```
i12 : invariants myAction

Warning: stopping condition not met!
Output may not generate the entire ring of invariants.
Increase value of DegreeBound.


               2    2
o12 = {x + y, x  + y }

o12 : List
```

不変式環の生成元として、`x+y`と`x^2+y^2`が求まった。何故かエラーが出て、これらで不変式環が生成されるかどうか分からないと書いてある。言われた通り`DegreeBound`を大きくしても、同じエラーが出るので気にしないことにする。生成元は正しく求まっているようだ。

演習

1. n=3,4,5について、n次対称群のn変数多項式環への置換作用に対する不変式環を求めよ。ヒント：n次対称群を生成するn-1個の互換に対応する行列を`apply(n-1, i -> permutationMatrix(n, [i+1,i+2]))`で作ることができる。
2. 対称群の代わりに交代群で同じ事をせよ。
3. 各不変式環の生成元の間の関係式を求めよ。

4. 上で求めた各不変式環について、コマンド `hilbertSeries` を用いてヒルベルト（ポアンカレ）級数を求めよ。

係数体の拡大

係数体を有理数体とすると、定義できる群作用が限られる。より多くの群作用を扱うためには、係数体を拡大する必要がある。特に、対角作用を扱うためには、1の冪根を体に添加する必要がある。

有理数体に1の3乗根を添加してみよう。

```
i1 : L = toField(QQ[a]/(a^2+a+1)) -- 環を体として扱う

o1 = L

o1 : PolynomialRing

i3 : a^3

o3 = 1

o3 : L
```

ここで、 `toField(QQ[a]/(a^3-1))` としないように注意すること。 `a^3-1` は既約多項式ではないので、体を定めないがエラーメッセージは出ない。しかし、後の計算でおかしな結果になる。正しい既約多項式を入力しよう。

新しく定義した体 `L` を係数体として、以下のように位数3の巡回群作用による不変式環を計算できる。

```
i20 : R = L[x,y]

o20 = R

o20 : PolynomialRing

i21 : myAction = finiteAction(matrix{{a,0},{0,a^2}},R)

o21 = R <- {| a 0    |}
            | 0 -a-1 |

o21 : FiniteGroupAction

i22 : invariantRing myAction

Warning: stopping condition not met!
Output may not generate the entire ring of invariants.
Increase value of DegreeBound.
```

```
o22 =          3   3
      L[x*y, y , x ]

o22 : RingOfInvariants
```

得られた不変式環が正規（integrally closed）であることを、コマンド`isNormal`を用いて確認できる。

```
i67 : T = invariantRing myAction;

Warning: stopping condition not met!
Output may not generate the entire ring of invariants.
Increase value of DegreeBound.

i69 : I = definingIdeal T

             3
o69 = ideal(u  - u u )
             1    2 3

o69 : Ideal of L[u ..u ]
                  1   3

i70 : T2 = (ring I)/I -- 不変式環を商環として表した

o70 = T2

o70 : QuotientRing

i72 : isNormal T2

o72 = true
```

`invariantRing`の出力として得られる`RingOfInvariants`クラスのインスタンスには、`isNormal`など適用する様々なコマンドが適用できない。上のようにして、`QuotientRing`クラスのインスタンスを作る必要がある。

演習

1. 巡回群による対角作用による不変式環を、考える変数の個数や巡回群の位数を様々に変化させ、計算せよ。
2. 各ケースにおいて、生成元の個数と関係式の個数がいくつになるか確認せよ。
3. 得られた不変式環は全て正規であることを確認せよ。

4. ヒルベルト・ポアンカレ級数を計算し、Molienの公式が成立することを確認せよ。
5. (Advanced) 不変式環がCohen-Macaulayかどうか、Gorensteinかどうかを確認してみよう。（ヒント：これらの性質を確認するのに必要なパッケージを探そう。）

実は対角作用の場合は、係数体を拡大しないでも不変式環を以下のように計算できる。（補足：スキーム論的には、群スキームμ_lの作用による不変式環だと解釈できる。係数体にかかわらず、同じ単項式が生成元となる。）

```
i31 : R = QQ[x,y]

o31 = R

o31 : PolynomialRing

i32 : A = diagonalAction(matrix{{1,2}},{3},R)

o32 = R <- ZZ/3 via

      | 1 2 |

o32 : DiagonalAction

i33 : invariantRing A

o33 =          3   3
      QQ[x*y, y , x ]

o33 : RingOfInvariants
```

商写像

不変式環から多項式環への埋め込み写像

$$k[x_1, \dots, x_n]^G \hookrightarrow k[x_1, \dots, x_n]$$

は、商写像

$$\mathbb{A}_k^n \to \mathbb{A}_k^n / G$$

に対応する。上で作った不変式環`T2`について、埋め込み写像をM2上で以下のように定める。

以下の計算は、有理数体上でしか実行できないので、もう一度有理数体上の不変式環を準備して商環として表す。

```
i112 : R = QQ[x,y]

o112 = R

o112 : PolynomialRing

i113 : A = diagonalAction(matrix{{1,2}},{3},R)

o113 = R <- ZZ/3 via

       | 1 2 |

o113 : DiagonalAction

i115 : I = definingIdeal invariantRing A

              3
o115 = ideal(u  - u u )
              1    2 3

o115 : Ideal of QQ[u ..u ]
                    1   3

i116 : T = (ring I)/I

o116 = T

o116 : QuotientRing

i117 : describe T

       QQ[u ..u ]
           1   3
o117 = ----------
         3
        u  - u u
         1    2 3
```

次に、TからRへの写像を変数の行き先を指定することで定める。

```
i120 : use R

o120 = R
```

```
o120 : PolynomialRing

i121 :  f = map(R,T,{x*y,y^3,x^3})

                          3   3
o121 = map (R, T, {x*y, y , x })

o121 : RingMap R <-- T

i122 : isWellDefined f

o122 = true
```

定義域が多項式環ではなく商環の場合、well-definedになる保証はないので、well-definedかどうか念のためにチェック。

次にアフィン平面$\mathbb{A}_k^2 = k^2$の点(1,2)を取り、fによる像、その逆像を計算する。

```
i127 : m1 = ideal(x-1,y-2)

o127 = ideal (x - 1, y - 2)

o127 : Ideal of R

i128 : m2 = preimage(f,m1)

o128 = ideal (u  - 2, u  - 1, u  - 8)
               1       3       2

o128 : Ideal of T

i129 : m3 = f(m2)

                         3       3
o129 = ideal (x*y - 2, x  - 1, y  - 8)

o129 : Ideal of R

i130 : decompose m3

                                                  2
o130 = {ideal (y - 2, x - 1), ideal (2x + y + 2, y  + 2y + 4)}

o130 : List
```

上の計算で、(1,2)の商写像による像はk^3の点として(2,8,1)であることが分かる。また、点(2,8,1)の商写像による逆像は、イデアル$(xy-2, x^3-1, y^3-8)$で与えられることが分かった。このイデアルは二つのminimal associated primes$(x-1, y-2), (2x+y+2, y^2+2y+4)$を持つ。二つの目のイデアルは有理数体上では、これ以上分解できないが。二点$(\zeta, \zeta^2 2), (\zeta^2, \zeta 2)$のペアに対応する。